Cálculo de primitivas

Los objetivos de esta práctica son:

- Manejar Maple para integrar funciones racionales, trigonométricas e irracionales cuadráticos.
- Aprender a usar la descomposición en fracciones simples (en diferentes cuerpos), la integración por partes y los cambios de variable.

Introducción

El teorema fundamental del cálculo integral asegur que cada función continua tiene una primitiva en un intervalo. También se sabe que dos primitivas en el mismo intervalo se diferencian en una constante. Sin embargo hay funciones continuas que no tienen *primitiva elemental.* Dos ejemplos:

```
> int(exp(-x^2), x);
```

$$\frac{1}{2}\sqrt{\pi}\ \mathrm{erf}(x)$$

```
> int(sqrt(x^3+8),x);
```

$$\frac{2}{5}x\sqrt{x^3+8}+\frac{48}{5}\frac{1}{\sqrt{x^3+8}}\Bigg((3-\mathrm{I}\sqrt{3})\sqrt{\frac{x+2}{3-\mathrm{I}\sqrt{3}}}\sqrt{\frac{x-1-\mathrm{I}\sqrt{3}}{-3-\mathrm{I}\sqrt{3}}}\sqrt{\frac{x-1+\mathrm{I}\sqrt{3}}{-3+\mathrm{I}\sqrt{3}}}\ \mathrm{EllipticF}\left(\sqrt{\frac{x+2}{3-\mathrm{I}\sqrt{3}}},\sqrt{\frac{-3+\mathrm{I}\sqrt{3}}{-3-\mathrm{I}\sqrt{3}}}\right)\Bigg)$$

Las funciones erf y EllipticF, que aparecen en los ejemplos anteriores, no son elementales.

Hay muchas funciones cuya primitiva es elemental. Entre ellas destacan las funciones racionales. De la misma forma el cálculo de primitivas de las funciones trigonométricas, irracionales cuadráticos y binomias se reduce al cálculo de primitivas de funciones racionales.

La orden **Int** permite escribir la integral, mientras que **int** la calcula. Veamos un ejemplo:

```
> Int(1/(x^2 - 6*x + 18),x);int(1/(x^2 - 6*x + 18), x);
```

$$\int \frac{1}{x^2-6x+18}\,dx$$

$$\frac{1}{3}\arctan\left(\frac{1}{3}x-1\right)$$

Como vemos, hay que indicar la función y cuál es la variable. También podemos calcular el valor mediante la orden **value**:

```
> value(Int(1/(x^2 - 6*x + 18), x));
```

$$\frac{1}{3}\arctan\left(\frac{1}{3}x-1\right)$$

En general y para lo que sigue es conveniente cargar el paquete **student.**

Descomposición en fracciones simples

Para descomponer un cociente de polinomios en fracciones simples en el cuerpo que queramos usamos la orden **convert**(expresión**,parfrac,x)**.

```
> restart: with(student):
> convert(x^3/(x^2+6*x+8),parfrac,x);
```

$$x-6+\frac{32}{x+4}-\frac{4}{x+2}$$

Si no indicamos nada más, Maple utiliza por defecto el mínimo cuerpo que contiene a los coeficientes. Otro ejemplo más complicado es este:

```
> f:=x->1/(x^4+16):convert(f(x),parfrac,x);
```

$$\frac{1}{x^4+16}$$

Parece que Maple no sabe hacer esta descomposición. El problema está en que factoriza el denominador en el mínimo cuerpo que contiene a los coeficientes, es decir, los racionales. Debemos sugerirle que use otro cuerpo. Para ello calculamos las raíces del polinomio:

```
> solve({x^4+16=0},x);
```

$$\{x=\sqrt{2}+\mathrm{I}\sqrt{2}\},\{x=-\sqrt{2}+\mathrm{I}\sqrt{2}\},\{x=-\sqrt{2}-\mathrm{I}\sqrt{2}\},\{x=\sqrt{2}-\mathrm{I}\sqrt{2}\}$$

Observamos que $\sqrt{2}$ aparece en la expresión de las soluciones (la unidad imaginaria también aparece, pero a nosotros no nos interesa factorizar con números complejos). Entonces utilicemos la orden **parfrac** de la siguiente forma:

```
> convert(f(x),parfrac,x,sqrt(2));
```

$$\frac{1}{32}\frac{-4+x\sqrt{2}}{-x^2+2x\sqrt{2}-4}+\frac{1}{32}\frac{4+x\sqrt{2}}{x^2+2x\sqrt{2}+4}$$

Cambios de variable

Para hacer un cambio de variable en una integral usamos **changevar(u=f(x),a,u)**, donde **a** representa la integral en la que hacemos el cambio, **u** es la nueva variable de integración y **u=f(x)** es la ecuación que define el cambio de variable.

```
> restart: with(student):
> a:=Int(1/(4*x + 17)^6,x);
```

$$a:=\int\frac{1}{(4x+17)^6}\,dx$$

Hacemos el cambio de variable $u=4x+17$:

```
> a1:=changevar(u=4*x+17,a,u);
```

$$a1:=\int\frac{1}{4u^6}\,du$$

Calculamos la integral:

```
> a2:=value(a1);
```

$$a2:=-\frac{1}{20u^5}$$

Deshacemos el cambio de variable (¡no hay que olvidarlo!) mediante la orden **subs**:

```
> Int(1/(4*x + 17)^6,x)=subs(u=4*x+17,a2)+C;
```

$$\int\frac{1}{(4x+17)^6}\,dx=-\frac{1}{20(4x+17)^5}+C$$

Integración por partes

Usaremos la orden **intparts(a,f(x))**, donde **a** representa la integral en la que queremos hacer la integración por partes y **f(x)** es la función que aparece sin derivar:

$$\int f(x)\, g'(x)\, dx = f(x)\, g(x) - \left(\int f'(x)\, g(x)\, dx\right).$$

```
> restart: with(student):
> a:=Int(x*log(x),x);
```

$$a := \int x \ln(x)\, dx$$

Hacemos integración por partes con $f(x) = \log(x)$:

```
> a1:=intparts(a,log(x));
```

$$a1 := \frac{1}{2} \ln(x)\, x^2 - \left(\int \frac{1}{2} x\, dx\right)$$

Calculamos la integral:

```
> a2:=value(a1);
```

$$a2 := \frac{1}{2} \ln(x)\, x^2 - \frac{1}{4} x^2$$

Y finalmente presentamos el resultado:

```
> Int(x*log(x),x)=a2+C;
```

$$\int x \ln(x)\, dx = \frac{1}{2} \ln(x)\, x^2 - \frac{1}{4} x^2 + C$$

Completar cuadrados

```
> restart; with(student):
```

En algunos tipos de integrales es útil completar cuadrados. Aunque esta orden no es exclusiva del cálculo de integrales, veamos cómo se puede usar en este ejemplo:

```
> a:=Int(1/(2*x^2+3*x+8),x);
```

$$a := \int \frac{1}{2x^2 + 3x + 8}\, dx$$

Completamos cuadrados en el denominador:

```
> completesquare(2*x^2+3*x+8,x);
```

$$2\left(x + \frac{3}{4}\right)^2 + \frac{55}{8}$$

Observando el resultado, hacemos el cambio de variable $t = \sqrt{\frac{16}{55}}\left(x + \frac{3}{4}\right)$:

```
> b:=changevar(t=sqrt(2*8/55)*(x+3/4),a,t);
```

$$b := \int \frac{1}{4} \frac{\sqrt{55}}{\frac{1}{440}\left(3\sqrt{55} - 55t\right)^2 - \frac{3}{220}\left(3\sqrt{55} - 55t\right)\sqrt{55} + 8}\, dt$$

Aparentemente el resultado es más complicado, pero se tiene que poder simplificar. Lo intentamos con alguna de las órdenes **simplify**, **expand** o **factor**:

```
> simplify(b);
```

$$\frac{2}{55}\sqrt{55}\left(\int \frac{1}{1+t^2}\, dt\right)$$

Esta última integral ya es inmediata:

```
> value(b);
```

$$\frac{2}{55}\sqrt{55}\arctan(t)$$

Finalmente, deshacemos el cambio de variable y escribimos el resultado:

```
> a=subs(t=sqrt(2*8/55)*(x+3/4),value(b))+C;
```

$$\int \frac{1}{2x^2 + 3x + 8}\, dx = \frac{2}{55}\sqrt{55}\arctan\left(\frac{4}{55}\sqrt{55}\left(x + \frac{3}{4}\right)\right) + C$$

Ejemplo: calcular las primitivas de la función $\frac{\sqrt{x}+1}{\left(\sqrt{x}-1\right)(x+2)}$.

```
> restart: with(student):
> a:=Int((sqrt(x)+1)/((sqrt(x)-1)*(x+2)),x);
```

$$a := \int \frac{\sqrt{x}+1}{\left(\sqrt{x}-1\right)(x+2)}\, dx$$

Podemos calcularla directamente:

```
> value(a);
```

$$\frac{4}{3}\ln\left(\sqrt{x}-1\right) + \frac{1}{3}\ln(x+2) + \frac{4}{3}\sqrt{2}\arctan\left(\frac{1}{2}\sqrt{x}\sqrt{2}\right)$$

Pero también nos interesa conocer algunos pasos intermedios.

Para empezar, hacemos el cambio de variable $t = \sqrt{x}$:

```
> changevar(t=sqrt(x),a,t);
```

$$\int \frac{2(t+1)t}{(t-1)\left(t^2+2\right)}\, dt$$

Descomponemos la función en fracciones simples:

```
> convert(2*((t+1)*t)/((t-1)*(t^2+2)), parfrac,t);
```

$$\frac{4}{3}\frac{1}{t-1} + \frac{2}{3}\frac{4+t}{t^2+2}$$

Calculamos por separado las primitivas de las dos fracciones simples:

```
> Int(4/3*1/(t-1),t)=int(4/3*1/(t-1),t);
  Int(2/3*(4+t)/(t^2+2),t)=int(2/3*(4+t)/(t^2+2),t);
```

$$\int \frac{4}{3}\frac{1}{t-1}\, dt = \frac{4}{3}\ln(t-1)$$

$$\int \frac{2}{3}\frac{4+t}{t^2+2}\, dt = \frac{1}{3}\ln\left(t^2+2\right) + \frac{4}{3}\sqrt{2}\arctan\left(\frac{1}{2}t\sqrt{2}\right)$$

Si en estas primitivas deshacemos el cambio $t = \sqrt{x}$ obtenemos el resultado inicial.